

Discrete Choice Modeling
William Greene
Stern School of Business, New York University

Lab Session 2
Binary Choice Modeling with Panel Data

This assignment will extend the models of binary choice and ordered choice to panel data frameworks. The exercises will use the health care data, `healthcare.csv`. Since these are a panel data set, we begin by identifying it as one

```
xtset id
```

1. Cluster correction in a nonlinear model

We start the exercise by noting that in fitting a model using a panel data set such as this one, conventional estimators such as the binary logit model may be consistent (that remains to be seen), but either way, the conventionally computed standard errors are likely to be biased downward. The ‘cluster correction’ is the usual device for dealing with this issue. We’ll use a probit model to investigate. It seems evident in the table that the correction actually does correct something. This is suggestive of the presence of common effects that are inducing correlation across observations.

```
*=====
* We display the diagonal elements of the covariance matrices.
*=====
gen logincome = log(income)
global x age educ logincome married hhkids hsat
*=====
* Comparing pooled vs. clustered standard errors
*=====
quietly probit doctor $x
estimates store pooled
matrix vh = e(V)
matrix vh = vecdiag(vh)
matrix vh = vh'
quietly probit doctor $x,cluster(id)
estimates store clustered
estimates table pooled clustered, b(%7.4f) se(%9.6f) t
matrix vc = e(V)
matrix vc = vecdiag(vc)
matrix vc = vc'
matrix list vh
matrix list vc
```

2. Pooled and Random Effects Models

In the random effects *linear regression* setting, the least squares estimator is consistent, though the conventional asymptotic covariance matrix, $s^2(\mathbf{X}'\mathbf{X})^{-1}$, is inappropriate. There is a misperception in some parts that the same is true for nonlinear models such as the probit model. In fact, when the model is nonlinear, even a random effect induces a bias in the conventional estimator, usually toward zero. Let’s find out.

```
probit doctor $x
margins,dydx($x)
xtprobit doctor $x, re
margins,dydx($x)
```

The displayed results seem to be consistent with the observation. To reiterate, just fixing the standard errors does not solve the problem caused by omitted common (random) effects.

3. Conditional and Unconditional Logit Fixed Effects Estimation.

For the binary logit model, the Chamberlain form of the fixed effects estimator is consistent while the unconditional (brute force) fixed effects estimator is inconsistent. (This is the *incidental parameters problem* that arises when T is small. In our unbalanced panel here, the largest group size is 7, and most groups have less than that. Thus, T is small here.) Fit the logit model by the two approaches, and compare the results. Are they very different? To see if we can't highlight the effect, let's look at the standard case, with T = 2. How different are the results now? Remember, in the T=2 case, $\text{plim } \mathbf{b}_{MLE} = 2\beta$ while $\text{plim } \mathbf{b}_C = \beta$. Do the results seem to bear this out? Note, going forward. There is no conditional fixed effects estimator for the probit model

```
*=====
* Comparing unconditional and conditional fixed effects estimators
*=====
quietly logit doctor income educ i.id if id <= 100
estimates store unconditional
quietly xtlogit doctor income educ if id <= 100, fe
estimates store conditional
estimates table unconditional conditional,b se t
*=====
* T=2 cases illustrates the 2*beta result
*=====
quietly logit doctor income educ i.id if id <= 500 & ti == 2
estimates store unconditional
quietly xtlogit doctor income educ if id <= 500 & ti == 2, fe
estimates store conditional
estimates table unconditional conditional,b se t
```

4. Test for Fixed Effects vs. No Effects in the Logit Model

In order to test for the need for fixed effects in the logit model, we can't use the likelihood ratio test because the unrestricted estimator is inconsistent. We can use a Hausman test, instead in the conditional logit model. (This was developed in full in Cicchetti's 1986 application of this model.) This uses the chi-squared statistic

$$H = (\mathbf{b}_C - \mathbf{b}_R)' [\mathbf{V}_C - \mathbf{V}_R]^{-1} (\mathbf{b}_C - \mathbf{b}_R)$$

where 'C' refers to the Chamberlain, conditional estimator and 'R' refers to the 'restricted' estimator which has only a single constant term. Note that \mathbf{b}_R is the subvector of the restricted estimator that strips off the overall constant term – it keeps only the slope coefficients. Using the model suggested in the commands below, carry out the test. What is the result? Do you reject the hypothesis? (What is the null hypothesis?)

```
global x income educ hhkids
logit doctor $x
matrix bp = e(b)
```

```

matrix bp = bp'
matrix bp = bp[1..3,1..1]
matrix vp = e(V)
matrix vp = vp[1..3,1..3]
xtlogit doctor $x,fe
matrix bf = e(b)
matrix bf = bf'
matrix vf = e(V)
matrix db = bf - bp
matrix dv = vf - vp
matrix htest = db' * invsym(dv) * db
matrix list htest

```

Note, it is not guaranteed that the difference matrix in the statistic is positive definite. To find out if it is, we will look at the characteristic roots. They must all be positive. Are they?

```

matrix symeigen x lambda = dv
matrix list lambda

```

5. Testing for Random Effects in the Probit Model

Testing for random effects in the probit model is straightforward using the likelihood ratio or Wald test. The LM test can also be used, but it has a peculiar feature that requires some special calculations. (See

http://web-docs.stern.nyu.edu/old_web/economics/docs/workingpapers/2012/Greene-LMTestsRandomEffects_Sept2012.pdf)

The LM test has an advantage over the other two in that it can be computed without fitting the random effects model. (Stata does not compute the LM test for the probit model.)

The following computes two test statistics, a Wald statistic and a likelihood ratio statistic. The second Probit command fits the full RE model then we obtain the likelihood ratio and Wald statistics for the tests.

```

*=====
* Testing for random effects
*=====
probit doctor $x
scalar llpool = e(ll)
xtprobit doctor $x, re
scalar llrem = e(ll)
*=====
* Note wald test with coefficient rho
* this is the Wald statistic based on the results
*=====
display (.393863/.0105553)^2
display 2*(llrem - llpool)

```

6. Mundlak's Approach – Correlated Random Effects

The fixed effects model holds that

$$y_{it}^* = \alpha_i + \beta'x_{it} + \varepsilon_{it}$$

and $E[\alpha_i|X_i] = g(X_i)$; that is, the effects are correlated with the other exogenous variables, X_i . Mundlak's suggestion is that the correlation can be accommodated through the group means (using the time varying variables),

$$\alpha_i = \alpha + \gamma' \bar{\mathbf{x}}_i + u_i$$

where u_i is uncorrelated with \mathbf{X}_i . Inserting the second equation into the first, we obtain what amounts to a random effects model;

$$y_{it}^* = \alpha + \beta' \mathbf{x}_{it} + \gamma' \bar{\mathbf{x}}_i + u_i + \varepsilon_{it}$$

This suggests a parsimonious treatment of the common effects in the model, and as well, a test of the presence of fixed effects. If the formulation above is (approximately) appropriate, then a test of the null hypothesis that $\gamma = 0$ would be a test of the null hypothesis of the random effects against the alternative of the fixed effects model. The following uses the Mundlak formulation to test for fixed effects. The commands involve some new features, so there is some annotation. The integration needed to compute random effects estimates is time consuming. We use Hermite integration.

```
bysort id: egen agebar=mean(age)
bysort id: egen educbar=mean(educ)
bysort id: egen logincbar=mean(logincome)
bysort id: egen marrbar=mean(married)
bysort id: egen kidsbar=mean(hhkids)
bysort id: egen hsatbar=mean(hsat)
global means agebar educbar logincbar marrbar kidsbar hsatbar
xtprobit doctor $x,re
scalar logl0=e(ll)
xtprobit doctor $x $means,re
test $means
scalar logl1=e(ll)
display 2*(logl1 - logl0)
```

7. Estimating the Random effects model as a Random Parameters Model.

The random effects model is a random parameters model with, in this case, only a random constant. (We will examine more elaborate specifications later in the course.) The RP formulation would be

$$y_{it}^* = (\alpha + \sigma w_i) + \beta' \mathbf{x}_{it} + \varepsilon_{it}.$$

Thus far, we have fit the REM using the Butler and Moffitt quadrature based method. We will now fit the model as an RP model using maximum simulated likelihood. We will then compare the results for the two methods. As you saw in parts 5 and 6, the Butler and Moffitt estimator reports RHO in the output – this equals the squared correlation between observations in a group. The simulation estimator reports SIGMA, the standard deviation of the common individual effect, w_i . To compare the two estimates of ρ , you must compute $\rho^* = \sigma^2 / (1 + \sigma^2)$ from the random parameters estimates. What do you find? Are the estimates of the other slopes nearly the same? The two log likelihoods are a bit different as well – they are two different approximations to the true log likelihood that includes the true but uncomputable values of the normal integrals. How close are they?

```
xtprobit doctor $x,re
meprobit doctor $x || id:
```

8. Fixed and Random Effects.

The disadvantage of the random effects estimator is that it requires an assumption that the individual effects are uncorrelated with the included variables. If that assumption is not met, the estimator is inconsistent. The fixed effects estimator is inconsistent when T is not large. Thus, both estimators have problems. Chamberlain's conditional estimator provides a way to estimate the logit fixed effects model consistently. One other complication is that the conditional estimator is only available for the logit model. We can fit a random effects logit model, but it is a bit peculiar in that the underlying utility function has a normally distributed random term and a logistically distributed common effect. The mixture of the two seems a bit peculiar. Nonetheless, we will use that model for this experiment.

**** Stata does not compute a true fixed effects logit model, so the comparison of the estimators to the unconditional FEM is omitted.

```
Namelist      ; x = income,educ,hhkids $
Logit         ; if[Ti = 7] ;lhs = doctor ; Rhs = x,one
              ; random ; panel ; hpt=8 ; Table = REM $
Logit         ; if[Ti = 7] ;lhs = hospital ; Rhs = x ; panel ; Table = FEMCond$
Logit         ; if[Ti = 7] ;lhs = hospital ; Rhs = x ; panel ; FEM ; Table = FEM $
Maketable     ; REM,FEM,FEMCond $
```

9. Dynamic Probit Model

The following computes the parameters of the dynamic probit model that we discussed in class. To do the estimation, we will take advantage of what we found in part 7, and use maximum simulated likelihood rather than quadrature to fit the model. This approach builds on the Mundlak model in part 7. Note the computation of the matrix of group means could be done a bit more compactly.

```
*=====
* Doctor Visits
*=====
by id,sort: gen dlag = doctor[_n-1]
by id,sort: gen d0 = doctor[1]
xtprobit doctor $x $means dlag d0,re
*=====
* public insurance
*=====
by id,sort: gen inslag = public[_n-1]
by id,sort: gen ins0 = public[1]
xtprobit public $x $means inslag ins0,re
```